NAG Library Routine Document

G13CEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

For a bivariate time series, G13CEF calculates the cross amplitude spectrum and squared coherency, together with lower and upper bounds from the univariate and bivariate (cross) spectra.

2 Specification

```
SUBROUTINE G13CEF (XG, YG, XYRG, XYIG, NG, STATS, CA, CALW, CAUP, T, SC,    &
                   SCLW, SCUP, IFAIL)

INTEGER            NG, IFAIL
REAL (KIND=nag_wp) XG(NG), YG(NG), XYRG(NG), XYIG(NG), STATS(4),            &
                   CA(NG), CALW(NG), CAUP(NG), T, SC(NG), SCLW(NG),         &
                   SCUP(NG)
```

3 Description

Estimates of the cross amplitude spectrum $A(\omega)$ and squared coherency $W(\omega)$ are calculated for each frequency ω as

$$A(\omega) = \left|f_{xy}(\omega)\right| = \sqrt{cf(\omega)^2 + qf(\omega)^2} \quad \text{and}$$

$$W(\omega) = \frac{\left|f_{xy}(\omega)\right|^2}{f_{xx}(\omega)f_{yy}(\omega)},$$

where

$cf(\omega)$ and $qf(\omega)$ are the co-spectrum and quadrature spectrum estimates between the series, i.e., the real and imaginary parts of the cross spectrum $f_{xy}(\omega)$ as obtained using G13CCF or G13CDF;

$f_{xx}(\omega)$ and $f_{yy}(\omega)$ are the univariate spectrum estimates for the two series as obtained using G13CAF or G13CBF.

The same type and amount of smoothing should be used for these estimates, and this is specified by the degrees of freedom and bandwidth values which are passed from the calls of G13CAF or G13CBF.

Upper and lower 95% confidence limits for the cross amplitude are given approximately by

$$A(\omega)\left[1 \pm \left(1.96/\sqrt{d}\right)\sqrt{W(\omega)^{-1} + 1}\right],$$

except that a negative lower limit is reset to 0.0, in which case the approximation is rather poor. You are therefore particularly recommended to compare the coherency estimate $W(\omega)$ with the critical value T derived from the upper 5% point of the F-distribution on $(2, d-2)$ degrees of freedom:

$$T = \frac{2F}{d - 2 + 2F},$$

where d is the degrees of freedom associated with the univariate spectrum estimates. The value of T is returned by the routine.

The hypothesis that the series are unrelated at frequency ω, i.e., that both the true cross amplitude and coherency are zero, may be rejected at the 5% level if $W(\omega) > T$. Tests at two frequencies separated by more than the bandwidth may be taken to be independent.

The confidence limits on $A(\omega)$ are strictly appropriate only at frequencies for which the coherency is significant. The same applies to the confidence limits on $W(\omega)$ which are however calculated at all frequencies using the approximation that $\operatorname{arctanh}\left(\sqrt{W(l)}\right)$ is Normal with variance $1/d$.

4 References

Bloomfield P (1976) *Fourier Analysis of Time Series: An Introduction* Wiley

Jenkins G M and Watts D G (1968) *Spectral Analysis and its Applications* Holden–Day

5 Arguments

1: XG(NG) – REAL (KIND=nag_wp) array *Input*

On entry: the NG univariate spectral estimates, $f_{xx}(\omega)$, for the x series.

2: YG(NG) – REAL (KIND=nag_wp) array *Input*

On entry: the NG univariate spectral estimates, $f_{yy}(\omega)$, for the y series.

3: XYRG(NG) – REAL (KIND=nag_wp) array *Input*

On entry: the real parts, $cf(\omega)$, of the NG bivariate spectral estimates for the x and y series. The x series leads the y series.

4: XYIG(NG) – REAL (KIND=nag_wp) array *Input*

On entry: the imaginary parts, $qf(\omega)$, of the NG bivariate spectral estimates for the x and y series. The x series leads the y series.

Note: the two univariate and the bivariate spectra must each have been calculated using the same method of smoothing. For rectangular, Bartlett, Tukey or Parzen smoothing windows, the same cut-off point of lag window and the same frequency division of the spectral estimates must be used. For the trapezium frequency smoothing window, the frequency width and the shape of the window and the frequency division of the spectral estimates must be the same. The spectral estimates and statistics must also be unlogged.

5: NG – INTEGER *Input*

On entry: the number of spectral estimates in each of the arrays XG, YG, XYRG and XYIG. It is also the number of cross amplitude spectral and squared coherency estimates.

Constraint: $\text{NG} \geq 1$.

6: STATS(4) – REAL (KIND=nag_wp) array *Input*

On entry: the four associated statistics for the univariate spectral estimates for the x and y series. STATS(1) contains the degrees of freedom, STATS(2) and STATS(3) contain the lower and upper bound multiplying factors respectively and STATS(4) contains the bandwidth.

Constraints:

$\text{STATS}(1) \geq 3.0$;
$0.0 < \text{STATS}(2) \leq 1.0$;
$\text{STATS}(3) \geq 1.0$.

7: CA(NG) – REAL (KIND=nag_wp) array *Output*

On exit: the NG cross amplitude spectral estimates $\hat{A}(\omega)$ at each frequency of ω.

8: CALW(NG) – REAL (KIND=nag_wp) array *Output*

On exit: the NG lower bounds for the NG cross amplitude spectral estimates.

9: CAUP(NG) – REAL (KIND=nag_wp) array *Output*

On exit: the NG upper bounds for the NG cross amplitude spectral estimates.

10: T – REAL (KIND=nag_wp) *Output*

On exit: the critical value for the significance of the squared coherency, T.

11: SC(NG) – REAL (KIND=nag_wp) array *Output*

On exit: the NG squared coherency estimates, $\hat{W}(\omega)$ at each frequency ω.

12: SCLW(NG) – REAL (KIND=nag_wp) array *Output*

On exit: the NG lower bounds for the NG squared coherency estimates.

13: SCUP(NG) – REAL (KIND=nag_wp) array *Output*

On exit: the NG upper bounds for the NG squared coherency estimates.

14: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, NG < 1,
or STATS(1) < 3.0,
or STATS(2) ≤ 0.0,
or STATS(2) > 1.0,
or STATS(3) < 1.0.

IFAIL $= 2$

A bivariate spectral estimate is zero. For this frequency the cross amplitude spectrum and squared coherency and their bounds are set to zero.

IFAIL $= 3$

A univariate spectral estimate is negative. For this frequency the cross amplitude spectrum and squared coherency and their bounds are set to zero.

IFAIL $= 4$

A univariate spectral estimate is zero. For this frequency the cross amplitude spectrum and squared coherency and their bounds are set to zero.

IFAIL $= 5$

A calculated value of the squared coherency exceeds 1.0. For this frequency the squared coherency is reset to 1.0 and this value for the squared coherency is used in the formulae for the calculation of bounds for both the cross amplitude spectrum and squared coherency. This has the consequence that both squared coherency bounds are 1.0.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

If more than one failure of the types 2, 3, 4 and 5 occurs then the failure type which occurred at lowest frequency is returned in IFAIL. However the actions indicated above are also carried out for failures at higher frequencies.

7 Accuracy

All computations are very stable and yield good accuracy.

8 Parallelism and Performance

G13CEF is not threaded in any implementation.

9 Further Comments

The time taken by G13CEF is approximately proportional to NG.

10 Example

This example reads the set of univariate spectrum statistics, the two univariate spectra and the cross spectrum at a frequency division of $\frac{2\pi}{20}$ for a pair of time series. It calls G13CEF to calculate the cross amplitude spectrum and squared coherency and their bounds and prints the results.

10.1 Program Text

```
    Program g13cefe

!     G13CEF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g13cef, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter              :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)              :: t
      Integer                         :: i, ifail, j, ng
```

```
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ca(:), calw(:), caup(:), sc(:),      &
                                          sclw(:), scup(:), xg(:), xyig(:),    &
                                          xyrg(:), yg(:)
      Real (Kind=nag_wp)               :: stats(4)
!     .. Executable Statements ..
      Write (nout,*) 'G13CEF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem size
      Read (nin,*) ng

!     Read in statistics
      Read (nin,*) stats(1:4)

      Allocate (xg(ng),yg(ng),xyrg(ng),xyig(ng),ca(ng),calw(ng),caup(ng),      &
        sc(ng),sclw(ng),scup(ng))

!     Read in data
      Read (nin,*)(xg(i),yg(i),xyrg(i),xyig(i),i=1,ng)

!     Calculate cross-amplitude spectrum
      ifail = -1
      Call g13cef(xg,yg,xyrg,xyig,ng,stats,ca,calw,caup,t,sc,sclw,scup,ifail)
      If (ifail/=0) Then
        If (ifail<2) Then
          Go To 100
        End If
      End If

!     Display results
      Write (nout,*) '      Cross amplitude spectrum'
      Write (nout,*)
      Write (nout,*) '                           Lower     Upper'
      Write (nout,*) '          Value     bound     bound'
      Write (nout,99999)(j-1,ca(j),calw(j),caup(j),j=1,ng)
      Write (nout,*)
      Write (nout,99998) 'Squared coherency test statistic =', t
      Write (nout,*)
      Write (nout,*) '         Squared coherency'
      Write (nout,*)
      Write (nout,*) '                           Lower     Upper'
      Write (nout,*) '          Value     bound     bound'
      Write (nout,99999)(j-1,sc(j),sclw(j),scup(j),j=1,ng)

100   Continue

99999 Format (1X,I5,3F10.4)
99998 Format (1X,A,F12.4)
    End Program g13cefe
```

10.2 Program Data

```
G13CEF Example Program Data
     9
 30.00000   .63858  1.78670   .33288
  2.03490 21.97712 -6.54995  0.00000
   .51554  3.29761   .34107 -1.19030
   .07640   .28782   .12335   .04087
   .01068   .02480  -.00514   .00842
   .00093   .00285  -.00033   .00032
   .00100   .00203  -.00039  -.00001
   .00076   .00125  -.00026   .00018
   .00037   .00107   .00011  -.00016
   .00021   .00191   .00007  0.00000
```

10.3 Program Results

```
 G13CEF Example Program Results

       Cross amplitude spectrum

                     Lower      Upper
          Value      bound      bound
    0    6.5499     3.9277    10.9228
    1    1.2382     0.7364     2.0820
    2    0.1299     0.0755     0.2236
    3    0.0099     0.0049     0.0197
    4    0.0005     0.0001     0.0017
    5    0.0004     0.0001     0.0015
    6    0.0003     0.0001     0.0010
    7    0.0002     0.0001     0.0007
    8    0.0001     0.0000     0.0018

 Squared coherency test statistic =      0.1926

          Squared coherency

                     Lower      Upper
          Value      bound      bound
    0    0.9593     0.9185     0.9799
    1    0.9018     0.8093     0.9507
    2    0.7679     0.5811     0.8790
    3    0.3674     0.1102     0.6177
    4    0.0797     0.0000     0.3253
    5    0.0750     0.0000     0.3182
    6    0.1053     0.0000     0.3610
    7    0.0952     0.0000     0.3475
    8    0.0122     0.0000     0.1912
```